How Does Decimation work?





Say you have an ADC with a limited 3bit resolution.
The ADC resolution defines a background RMS noise level.
The sample rate also defines the bandwidth for this RMS noise level.
So a 8msec sample rate creates a noise floor of 5.0mvolts.

Now use oversampling to add more bits. Increase the sample rate by four.



The oversampling does not change the rms value of the quantization error.
But it can move most quantization noise above the signal bandwidth.



So maybe now an 2.7m noise floor is good enough ENOB in the critical bandwidth.
But the signal is being run at a clock rate four times higher than needed.
And that is drawing supply current.

$$\mathrm{ENOB} = \frac{\mathrm{SNR} - 1.76\mathrm{dB}}{6.02\mathrm{dB}}.$$



It turns out that when the signal gets digitally filtered,
it adds resolution to the data.
In this case, a six stage FIR average filter is being used.
(For simplicity sake)



The running average is being done at the 8msec sample rate.
But since six stages are being used,
the lowpass output is seeing a factor of six increase in resolution.



The filtered spectrum will have a six time lower bandwidth.
But at the lower frequencies, the 2.6m noise floor remains the same.

Re-Sampling the output of the filter back to a 8msec rate,
is just sampling every fourth data point of the 2msec rate signal.
Most of the noise and the signal will be well under nyquist,
so the spectrum in the critical bandwidth will remain the same.

The signal will in fact have picked up more voltage resolution.
But what happens to the noise floor is what really is important.
From the ENOB standpoint, one bit of resolution has been added.

$$\mathrm{ENOB} = \frac{\mathrm{SNR} - 1.76\mathrm{dB}}{6.02\mathrm{dB}}.$$

===================MacSpiceCode==

```
Simple_Decimation
*        VIN
*
*    _|____________|_  _______|_      THO
*   / \           |   |    / \  
*  //\ \          | |\  _(   )    TH1  |\
*  \  \//         |_| >_| \_/   |_|\_  | >__________
*   \__/            |/  |  OFF0 |   >  |/    \   |\
*    _|_                |      _|/  OFF1      \  | \    DAC   \__|VOUTA|
*    ///              |D0|    |D1|     |D2|___|  |  /
*                      |        |______________|  |/
*                      |__________________________|
*
*
*   VOUTA     IN2
*          |\    /        IN3  |\
*   |IN|_|1\ __/               |2\___|OUT|
*          |/ S1   ____|_______|/
*          |/        _|_       |/
*               C1   ___
*                    ///
*
* |OUT1|_|1/z|___|1/z|___|1/z|_....
*                |         |
*              |a1|      |a2|     NORMAL FIR FORMAT
*                |         |
*                |         |___\ /‾‾\
*                |             // \_  \___|OUT2|
*                |_____________\\ /_  /
*                              / \__/
*.............................................................
```

```
*

*         F1        F2       F3        F4
* ____    __|__ V1  __|__ V2 __|__ V3  __|__  V4
*|OUT1|__|D_LAY|___|D_LAY|__|D_LAY|___|D_LAY|_...
*|____|  |_____|   |_____|  |_____|   |_____|
*          |         |        |         |
*          |_________|________|_________|
*                   |
*                   |  ____   DC VOLTAGE CONTROLED FORMAT
*       _/\  /\  /\_|_|OUT2|
*      _|_  \/  \/    |____|
*      ///   R1 = 1 Ohm


*=========Create_Signal=================
*V_SIN#   NODE_P NODE_N DC     VALUE  SIN(   V_DC   AC_MAG FREQ   DELAY  FDamp)
VIN       VIN    0      DC     0      SIN(   .5       .5     4                 )
*V_PULSE# NODE_P NODE_N DC     VALUE  PULSE( VINIT  VPULSE TDELAY TRISE  TFALL  PWIDTH PERIOD )
VCLKL     CLK    0      DC     0      PULSE( 0      1      1n     1n     1n     1m     2m     )

XPOSE1    CLK    CNTL   POS_E
XS_H1     VOUTA  CNTL   OUT1      SH

BTH0      D0     0      V =    u( V(VIN) -1/2)
BOFF0     VIN    OFF0   V =    V(D0)/2
BTH1      D1     0      V =    u( V(OFF0) -1/4)
BOFF1     OFF0   OFF1   V =    V(D1)/4
BTH2      D2     0      V =    u( V(OFF1) -1/8)
BOFF2     OFF1   OFF2   V =    V(D2)/8
BTH3      D3     0      V =    u( V(OFF2) -1/16)
BOFF3     OFF2   OFF3   V =    V(D3)/16
BTH4      D4     0      V =    u( V(OFF3) -1/32)
BDAC      VOUTA  0      V =    (V(D0)/2+V(D1)/4+V(D2)/8)

VF1       F1     0      DC     0.1666
XS_DLAY1  OUT1   V1     OUT2   CLK  F1   S_DLAY
XS_DLAY2  V1     V2     OUT2   CLK  F1   S_DLAY
XS_DLAY3  V2     V3     OUT2   CLK  F1   S_DLAY
XS_DLAY4  V3     V4     OUT2   CLK  F1   S_DLAY
XS_DLAY5  V4     V5     OUT2   CLK  F1   S_DLAY
XS_DLAY6  V5     V6     OUT2   CLK  F1   S_DLAY
ROUT2     OUT2   0    1

.control
*TRAN     TSTEP  TSTOP  TSTART TMAX   ?UIC?
tran      .05m    1      0      .05m
set       pensize = 2
plot      vin  vouta out1
plot      vin  vouta out2

echo "==================FFT_and_Plot====================================="
linearize

let       FFT_BandWidth_Hz = 10k
let       FFT_resolution_Hz = 1
echo     "FFT_BandWidth_Hz= $&FFT_BandWidth_Hz"
echo     "FFT_resolution_Hz= $&FFT_resolution_Hz"
set       specwindow = "hanning"
spec      $&FFT_resolution_Hz $&FFT_BandWidth_Hz $&FFT_resolution_Hz v(out2)
let       BW = 1/2m
let       gain = 1/(1 + frequency/(0.5*BW))
let       gain = gain*((.5/8)/sqrt(BW))
plot      mag (out2) gain loglog
echo "==================Done=============================================="

.endc

.MODEL    SW      SW(    VT=.5 VH=.1  RON=1 ROFF=100MEG)
*=========Sample_Hold=========================================
*
*            IN2
*     ___ |\    /          IN3 |\     ___
*   |IN  |_|1\__/  ____________|2\____|OUT|
*   |___|  | / S1   _|_        | /    |___|
*          |/       ___        |/
*                 C1 |
*                   _|_
*                   ///
.SUBCKT   SH     IN     CNTL      OUT
 B1       IN2    0      V =       v(IN )
 S1       IN2    IN3    CNTL     0    SW
 C1       IN3    0      .1u
 R1       IN3    0      100Meg
```

```
 B2        OUT    0      V =       v(IN3 )
.ENDS      SH

*=========POS_Edge==========================================
*
*   ___   |\             VBF  ____                      ___
*  |IN |__|3\_________________|    \   ____    ____  |   \   ____
*  |___|  | /  |          VLP|  PE \_|OUT|   |IN |_|PE \_|OUT|
*         |/  |/\  /\  /\____|     /  |___|   |___| |   / |___|
*                \/  \/    |   |___/                |__/
*              RLP        _|_
*
*                  CLP    ___
*                          |
*                         _|_
*                         ///
*
.SUBCKT   POS_E  IN     OUT
 BBUF     VBF    0      V =   u( v(IN )-.5 )
 RLP      VBF    VLP    10k
 CLP      VLP    0      1n   IC=0
 BAND     OUT    0      V =    u( u(v(VBF )-.5)*u(.5 -v(VLP ) ) -.1)
.ENDS     POS_E
*=========SCALED_DELAY============================================
*            CLK
*
*   ___      __|__                    ____
*  |VIN|____|D_LAY|_____________|VOUT|
*  |___|    |_____| |             |____|
*                   |      | F1
*                   | |\  V       ____
*                   |_|1\/\/\___|IOUT|
*                     |  /\/\/    |____|
*                     |/
*
.SUBCKT   S_DLAY VIN    VOUT  IOUT  CLK  F1
 XZ_DLAY  VIN    VOUT   CLK  Z_DLAY
 BOUT     IOUT   0      I  = -V(VOUT)*V(F1)
.ENDS     S_DLAY
*=========Z_DLAY=========================================
*                                ________________
*                               |   ___________  |
*   ___   |\          VBF      |  |   ____     | |  ____
*  |CLK|__|1\_________________|_/|\_\    \    | |_|    \
*  |___|  | /  |          VLP|    |NE \_  |    | PE \_
*         |/  |/\  /\  /\______|___|  2 / |  |___| 3   /  |
*                \/  \/   _|_      /___/  |      |___/    |
*              RLP        ___                              |
*                         _|_
*                  CLP ///
*
*  ___    |\  V1   /   V2        |\  V3   /   V4      |\   ___
* |IN |__|4\____/ ____________|5\____/ ___________|6\_|OUT|
* |___|  | / S1  NE  |  _|_    | / S2  PE  |  _|_   | /  |___|
*        |/          |  ___    |/          |  ___   |/
*       Rshunt1      |   | C1  Rshunt2     |   |C2
*     _/\  /\  /\__|  _|_   _/\  /\  /\__|  _|_
*    _|_  \/  \/       ///  _|_  \/  \/        ///
*    ///                    ///
.SUBCKT   Z_DLAY VIN    OUT    CLK
 B1       VBF    0      V =    u( v(CLK )-.5 )
 RLP      VBF    VLP    10k
 CLP      VLP    0      50n    IC=0
 BNOR2    NE     0      V =    1-u( u(v(VBF )-.5)+u(.5 -v(VLP ) ) -.1)
 BAND3    PE     0      V =      u( u(v(VBF )-.5)*u(.5 -v(VLP ) ) -.1)
 B4       V1     0      V =    V(VIN)
 S1       V1     V2     NE     0    SW
 R1       V2     0      100Meg
 C1       V2     0      1u
 B5       V3     0      V =    V(V2)
 S2       V3     V4     PE     0    SW
 R2       V4     0      100Meg
 C2       V4     0      100u
 B6       OUT    0      V =    V(V4)
.ENDS     Z_DLAY

.end


7.22.11_1.28PM
dsauersanjose@aol.com
Don Sauer
```